Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
NotebookOpen[wdir <> "\\MakeVSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 4; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
  Collect[Normal@Series[ℰ /. T → e^(ℏ t/2), {ℏ, 0, $TℏD}], ℏ, Together]]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
NonCommutativeMultiply[x_] := x;
NonCommutativeMultiply[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
       CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
   ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) ⧴ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]];
  S_i_[u_U] := DeleteCases[u, _i] ** NonCommutativeMultiply @@
     U_i[S_i /@ Reverse@Cases[u, x_i ⧴ x]];
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) :> (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U :> m[u]]; )
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.71875, {(28 t^2 γ^4 + 116 t γ^5 ϵ + 120 γ^6 ϵ^2) CU[y, y, y, x, x] +
   <<23>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q=e^{\hbar\gamma\epsilon}$, $A=e^{-\hbar\epsilon a}$, $T=e^{\hbar t/2}$, and $[f,g]_q:=fg-qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar}=\langle t,y,a,x\rangle/\mathcal{R}$, where $\mathcal{R}=\left([t,*]=0,\ [a,y]=-\gamma y,\ [x,y]_q=\left(1-T^2A^2\right)/\hbar,\ [x,a]=-\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ℏ)]; (*T=SS[e^(ℏ t/2)];*)
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] = (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 e^(-2 ϵ a ℏ)) / ℏ], {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\left\{\left\{\{QU[y], QU[y]\} \to 0,\ \{QU[y], QU[a]\} \to \gamma\, QU[y],\ \{QU[y], QU[x]\} \to \left(t + \frac{t^2 \hbar}{2} + \frac{t^3 \hbar^2}{6} + \frac{t^4 \hbar^3}{24}\right) QU[\,] + \left(-2 \epsilon - 2 t \epsilon \hbar - t^2 \epsilon \hbar^2 - \frac{1}{3} t^3 \epsilon \hbar^3\right) QU[a] + \left(2 \epsilon^2 \hbar + 2 t \epsilon^2 \hbar^2 + t^2 \epsilon^2 \hbar^3\right) QU[a, a] + \left(-\gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x]\right\},\right.$$
$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y],\ \{QU[a], QU[a]\} \to 0,\ \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$
$$\left\{\{QU[x], QU[y]\} \to \left(-t - \frac{t^2 \hbar}{2} - \frac{t^3 \hbar^2}{6} - \frac{t^4 \hbar^3}{24}\right) QU[\,] + \left(2 \epsilon + 2 t \epsilon \hbar + t^2 \epsilon \hbar^2 + \frac{1}{3} t^3 \epsilon \hbar^3\right) QU[a] + \left(-2 \epsilon^2 \hbar - 2 t \epsilon^2 \hbar^2 - t^2 \epsilon^2 \hbar^3\right) QU[a, a] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x],\right.$$
$$\left.\left.\{QU[x], QU[a]\} \to -\gamma\, QU[x],\ \{QU[x], QU[x]\} \to 0\right\}\right\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs @ \$T$\hbar$D=5, \$TϵD=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp]
  }] // Timing
```

```
{50.1406, {<<1>>, 0}}
```

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs @ \$T$\hbar$D=5, \$TϵD=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → e^(ℏ t/2)}, ℏ → 0] - lhs] // HL
  }] // Timing
```

$$\{35.875,\ \{2 \left(8 t^2 \gamma^4 + 16 t \gamma^5 \epsilon\right) CU[y, y, y, x, x] + \ll 107 \gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x],$$
$$2 \left(8 t \gamma^6 \epsilon^2 \hbar + 12 t^2 \gamma^6 \epsilon^2 \hbar^2 + \frac{28}{3} t^3 \gamma^6 \epsilon^2 \hbar^3\right) QU[y, y, y, x, x] + \ll 566 \gg + \left(\gamma \epsilon \hbar + \frac{15}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, y, \ll 9 \gg, x, x],\ 0\}\}$$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 e^(ħ ϵ a) x], {a, x}],
  a → -QU@a, x → 𝕆_QU[SS[-T^-1 e^(ħ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → −CU[x] → CU[y], CU[a] → −CU[a] → CU[a], CU[x] → −CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\Big\{QU[y] \to \left(-1+\frac{t\hbar}{2}-\frac{t^2\hbar^2}{8}+\frac{t^3\hbar^3}{48}\right)QU[x]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3\right)QU[a,x]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3\right)QU[a,a,x]\to QU[y],\ QU[a]\to -QU[a]\to QU[a],$$

$$QU[x]\to\left(-1+\frac{t\hbar}{2}+\gamma\epsilon\hbar-\frac{t^2\hbar^2}{8}-\frac{1}{2}t\gamma\epsilon\hbar^2-\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{t^3\hbar^3}{48}+\frac{1}{8}t^2\gamma\epsilon\hbar^3+\frac{1}{4}t\gamma^2\epsilon^2\hbar^3\right)QU[y]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2+\gamma\epsilon^2\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3-\frac{1}{2}t\gamma\epsilon^2\hbar^3\right)QU[y,a]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3\right)QU[y,a,a]\to QU[x]\Big\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

```
AD$f = γ/ℏ e^(ℏ (t/2 - (a+γ) ϵ)) ((Cosh[ℏ (a ϵ + γ ϵ/2 - t/2)] - Cosh[ℏ Sqrt[((t - γ ϵ)/2)^2 + ϵ ω]]) /
    (Sinh[γ ϵ ℏ/2] (a^2 ϵ + a γ ϵ - a t - ω)));
```

```
SS[AD$f] // Simplify
```

$$\frac{1}{24}\left(24+12\,(t-2\,(a+\gamma)\,\epsilon)\,\hbar+2\left(2\,t^2-7\,t\,(a+\gamma)\,\epsilon+\epsilon\left(7\,a^2\,\epsilon+13\,a\,\gamma\,\epsilon+6\,\gamma^2\,\epsilon+\omega\right)\right)\hbar^2+\left(t^3-5\,t^2\,(a+\gamma)\,\epsilon-2\,(a+\gamma)\,\epsilon^2\,\omega+t\,\epsilon\left(9\,a^2\,\epsilon+17\,a\,\gamma\,\epsilon+8\,\gamma^2\,\epsilon+\omega\right)\right)\hbar^3\right)$$

Scaling behaviour of AD$f:

```
Simplify[AD$f == ((AD$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

```
True
```

```
FullSimplify[
 AD$f == ((AD$f /. γ → 1) /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

```
True
```

ADeq

```
AD$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[AD, QU → CU,
  {a → CU@a, x → CU@x, y → 𝕊CU[SS[AD$f], a → CU[a], ω → AD$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[AD[z1 ** z2] - AD[z1] ** AD[z2]]], {z1, bas}, {z2, bas}]]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

```
SD$g = Sqrt[((Cosh[ℏ/2 Sqrt[t^2 + γ^2 ϵ^2 + 4 ϵ ϖ]] - Cosh[ℏ/2 (t - (2 a + γ) ϵ)]) /
    (Sinh[γ ϵ ℏ/2] (t (2 a + γ) - 2 a (a + γ) ϵ + 2 ϖ) ℏ / (2 γ)))];
```

$$\left\{\mathbb{SD}\$P = \frac{\text{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\, a\right)\right]-\text{Cosh}\left[\hbar\sqrt{\frac{t^2+\epsilon^2}{4}+\epsilon\,\varpi}\right]}{\hbar\,\text{Sinh}\left[\frac{-\epsilon\hbar}{2}\right]\left(\varpi-\epsilon\, a^2+(t-\epsilon)\, a+t/2\right)},\right.$$

```
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, ∊ → ∊ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
     SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 Simplify[SD$g == FullSimplify[
     √SD$Q /. c → a + 1/2 /. {ħ → γ^2 ħ, ∊ → ∊ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\text{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}\left(-t+\epsilon\right)\right)\hbar\right]-\text{Cosh}\left[\sqrt{\frac{1}{4}\left(t^2+\epsilon^2\right)+\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\left(t-\epsilon\right)-a^2\epsilon+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},\right.$$

$$\left.-\left(\left(4\left(\text{Cosh}\left[\frac{1}{2}\left(t-2\,c\,\epsilon\right)\hbar\right]-\text{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\epsilon+4\,\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True}\right\}$$

```
FullSimplify[SD$g /. {ħ → γ^2 ħ, ∊ → ∊ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]
```

$$\sqrt{2}\sqrt{\left(\left(\gamma^2\left(-\text{Cosh}\left[\frac{1}{2}\left(t-\left(2\,a+\gamma^2\right)\epsilon\right)\hbar\right]+\text{Cosh}\left[\frac{1}{2}\sqrt{t^2+\gamma^4\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{1}{2}\gamma^2\epsilon\,\hbar\right]\right)\Big/\left(\left(t\left(2\,a+\gamma^2\right)-2\,a\left(a+\gamma^2\right)\epsilon+2\,\varpi\right)\hbar\right)\right)}$$

```
SS[SD$g]
```

$$1+\frac{1}{48}\left(t^2-2\,a\,t\,\epsilon-t\,\gamma\,\epsilon+2\,a^2\epsilon^2+2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2$$

SDeq

```
SD$f = FullSimplify[e^(ħ (t/2-∊ a)) (SD$g /. {a → -a, t → -t})];
```

```
SS[SD$f]
```

$$1+\frac{1}{2}\left(t-2\,a\,\epsilon\right)\hbar+\frac{1}{48}\left(7\,t^2-26\,a\,t\,\epsilon+t\,\gamma\,\epsilon+26\,a^2\epsilon^2-2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2+$$
$$\frac{1}{96}\left(3\,t^3-16\,a\,t^2\epsilon+t^2\gamma\,\epsilon+30\,a^2\,t\,\epsilon^2-4\,a\,t\,\gamma\,\epsilon^2+2\,t\,\epsilon\,\varpi-4\,a\,\epsilon^2\varpi\right)\hbar^3$$

SDeq

```
SD$ϖ = γ CU[y, x] + ∊ CU[a, a] - (t - γ ∊) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → CU@a,
  x → S_CU[SS[SD$f], a → CU[a], ϖ → SD$ϖ] ** CU@x,
  y → S_CU[SS[SD$g] , a → CU[a], ϖ → SD$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

R in QU.

Quesne's formula:

Quesne

$$\mathbb{e}_{q\_,n\_}[x\_] := \text{Exp}\left[\sum_{k=1}^{n} \frac{(1-q)^k}{k\,(1-q^k)} x^k\right];\ \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$T\epsilon D}[x]$$

$$\text{Table}[\text{Together@SeriesCoefficient}[\mathbb{e}_{\rho,5}[x],\ \{x, 0, n\}],\ \{n, 0, 5\}]$$

$$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)}, 1\Big/\left((1+\rho)^2(1+\rho^2)(1+\rho+\rho^2)(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

$$\text{Table}[\text{HL@FunctionExpand}[\text{QFactorial}[n, \rho]\ \text{SeriesCoefficient}[\mathbb{e}_{\rho,5}[x],\ \{x, 0, n\}]],\ \{n, 0, 5\}]$$

```
{1, 1, 1, 1, 1, 1}
```

R

$$\text{QU}[R_{i\_,j\_}] := \mathbb{O}_{QU}\left[\text{SS}\left[\mathbb{e}^{\hbar\, b_1 a_2}\, \mathbb{e}_q[\hbar\, y_1 x_2]\ /.\ b_1 \to \gamma^{-1}(\epsilon\, a_1 - t_i)\right],\ \{y_1, a_1\} \to i,\ \{a_2, x_2\} \to j\right]$$

```
QU[R3,4] // Short
```

$$\text{QU}[\,] + \frac{\epsilon\,\hbar\,\text{QU}[a_3, a_4]}{\gamma} + \hbar\,\text{QU}[y_3, x_4] + \ll 19 \gg + \frac{\ll 1 \gg}{2 \ll 1 \gg} + \frac{\hbar^3\,\text{QU}[y_3, a_4, \ll 1 \gg, x_4]\, t_3^2}{2\gamma^2} - \frac{\hbar^3\,\text{QU}[a_4, a_4, a_4]\, t_3^3}{6\gamma^3}$$

Verifying R3 (~156 secs @ \$T$\hbar$D=4, \$TϵD=2):

```
{Short[lhs = QU[R1,2 ** R1,3 ** R2,3]], HL@Simp[lhs - QU[R2,3 ** R1,3 ** R1,2]]} // Timing
```

$$\left\{155.563, \left\{\text{QU}[\,] + \frac{\epsilon\,\hbar\,\text{QU}[a_1, a_2]}{\gamma} + \ll 1239 \gg + \text{QU}[y_1, x_3]\left(-\hbar^2 t_2 - \frac{1}{2}\hbar^3 t_2^2 - \frac{1}{6}\hbar^4 t_2^3\right) + \text{QU}[y_1, a_3, x_3]\left(\frac{\hbar^3 t_2^2}{\gamma} + \frac{\hbar^4 t_2^3}{2\gamma}\right), 0\right\}\right\}$$

$\mathbb{E}$

$\mathbb{E}[\omega, L, Q, P]$ means $\omega e^{L+Q} P$, where ω is a scalar, L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial.